



## Ejercicios Vectores y Matrices

### Ejercicio 1

Realizar un programa que defina un vector llamado "vector\_numeros" de 10 enteros, a continuación lo inicialice con valores aleatorios (del 1 al 10) y posteriormente muestre en pantalla cada elemento del vector junto con su cuadrado y su cubo.

Algoritmo VectorCuadradoCubo

Definir vector\_numeros Como Entero;

Dimension vector\_numeros[10]; // Declaramos el vector de 10 elementos

Definir i Como Entero;

// Inicializar el vector con valores aleatorios del 1 al 10

Para i <- 1 Hasta 10 Hacer

    vector\_numeros[i] <- Aleatorio(1,10);

FinPara

// Mostrar cada número con su cuadrado y su cubo

Escribir "Número | Cuadrado | Cubo";


Escribir "-----";

Para i <- 1 Hasta 10 Hacer

    Escribir vector\_numeros[i], " | ", vector\_numeros[i] \* vector\_numeros[i], " | ",  
vector\_numeros[i] \* vector\_numeros[i] \* vector\_numeros[i];

FinPara

FinAlgoritmo



## Ejercicio 2

Crear un vector de 5 elementos de cadenas de caracteres, inicializa el vector con datos leídos por el teclado. Copia los elementos del vector en otro vector pero en orden inverso, y muéstralo por la pantalla.

Algoritmo InvertirVectorCadenas

Definir vector\_original, vector\_inverso Como Cadena;

Dimension vector\_original[5]; // Declaramos los vectores

Dimension vector\_inverso[5];

Definir i Como Entero;

// Pedir al usuario que ingrese los 5 elementos del vector


Escribir "Ingrese 5 palabras o frases:";

```
Para i <- 1 Hasta 5 Hacer
    Escribir "Ingrese el elemento ", i, ": ";
    Leer vector_original[i];
FinPara

// Copiar los elementos en orden inverso
Para i <- 1 Hasta 5 Hacer
    vector_inverso[i] <- vector_original[6-i];
FinPara

// Mostrar el vector original
Escribir "Vector original:";
Para i <- 1 Hasta 5 Hacer
    Escribir vector_original[i];
FinPara

// Mostrar el vector invertido
Escribir "Vector en orden inverso:";
Para i <- 1 Hasta 5 Hacer
    Escribir vector_inverso[i];
FinPara
FinAlgoritmo
```



### Ejercicio 3

Diseñar el algoritmo correspondiente a un programa, que:

Crea una tabla bidimensional de longitud 5x5 y nombre 'matriz'.

Carga la tabla con valores numéricos enteros.

Suma todos los elementos de cada fila y todos los elementos de cada columna visualizando los resultados en pantalla.

Algoritmo SumarFilasColumnasMatriz

Definir matriz Como Entero;

Dimension matriz[5,5]; // Definimos una matriz 5x5

Definir suma\_fila, suma\_columna Como Entero;

Definir i, j Como Entero;

```

// Cargar la matriz con valores numéricos
Escribir "Ingrese los elementos de la matriz 5x5:";

Para i <- 1 Hasta 5 Hacer
    Para j <- 1 Hasta 5 Hacer
        Escribir "Ingrese el elemento [", i, ",", j, "]: ";
        Leer matriz[i, j];
    FinPara
FinPara

// Sumar los elementos de cada fila y cada columna
Escribir "Suma de los elementos de cada fila:";

Para i <- 1 Hasta 5 Hacer
    suma_fila <- 0;
    Para j <- 1 Hasta 5 Hacer
        suma_fila <- suma_fila + matriz[i, j]; // Sumar los elementos de la fila
    FinPara
    Escribir "La suma de la fila ", i, " es: ", suma_fila;
FinPara

Escribir "Suma de los elementos de cada columna:";

Para j <- 1 Hasta 5 Hacer
    suma_columna <- 0;
    Para i <- 1 Hasta 5 Hacer
        suma_columna <- suma_columna + matriz[i, j]; // Sumar los elementos de la columna
    FinPara
    Escribir "La suma de la columna ", j, " es: ", suma_columna;
FinPara
FinAlgoritmo

```

